Filtering discrete signals

signals

finite data structure specifies leftmost index and length

In[1]:=
```
signal[data_, left_] := Module[{length},
  length = Length[data];
  {data, left, length}
 ]
```

procedure for plotting a (finite section of a) signal

In[2]:=
```
showsignal[sig_] := Module[{data, left, length},
  {data, left, length} = sig;
  ListPlot[ data, Filling → Axis, PlotRange → All,
   DataRange -> {left, left - length - 1}, PlotStyle → PointSize[.015]]
 ]
```

example signal

In[3]:=
```
B[n_] := (3 / 16) * (4 / 5) ^Abs[n] + (1 / 20) * (-4 / 5) ^Abs[n]
```

In[4]:=
```
mysignal[n_] := signal[Table[B[k], {k, -n, n}], -n]
```

plotting the example signal

In[5]:=
```
showsignal[mysignal[20]]
```

Out[5]=



a finite signal to be used as low-pass filter

In[6]:=
```
lfilter = signal[{1 / 4, 1 / 2, 1 / 4}, -1]
```
Out[6]=
$$\left\{\left\{\frac{1}{4}, \frac{1}{2}, \frac{1}{4}\right\}, -1, 3\right\}$$

a finite signal to be used as high-pass filter

In[7]:=
```
hfilter = signal[{-1 / 4, 1 / 2, -1 / 4}, -1]
```
Out[7]=
$$\left\{\left\{-\frac{1}{4}, \frac{1}{2}, -\frac{1}{4}\right\}, -1, 3\right\}$$

convolution

In[8]:=
```
ListConvolve[{a, b}, {x, y, z}]
```
Out[8]=
```
{b x + a y, b y + a z}
```

Mathematica's list convolution for finite lists

In[9]:=
```
ListConvolve[{a, b, c}, {p, q, r, s}, {1, -1}, 0]
```
Out[9]=
```
{a p, b p + a q, c p + b q + a r, c q + b r + a s, c r + b s, c s}
```

convolution of finite signals is the same as multiplication of polynomials (z-powers with possibly negative exponents)

In[10]:=
$$(a + b z + c z^2)\,(p z^{-2} + q z^{-1} + r z^0 + s z^1)$$
Out[10]=
$$\left(r + \frac{p}{z^2} + \frac{q}{z} + s z\right)\,(a + b z + c z^2)$$

In[11]:=
```
Collect[Expand[%], z]
```
Out[11]=
$$c p + b q + a r + \frac{a p}{z^2} + \frac{b p + a q}{z} + (c q + b r + a s)\, z + (c r + b s)\, z^2 + c s z^3$$

filtering

a convolution procedure which takes care of the start- and stop-indices of finite signals

In[12]:=
```
convolve[sig1_, sig2_] :=
 Module[{data1, data2, left1, left2, length1, length2, sig, left, length},
  {data1, left1, length1} = sig1;
  {data2, left2, length2} = sig2;
  sig = ListConvolve[data1, data2, {1, -1}, 0];
  left = left1 + left2;
  length = length1 + length2 - 1;
  {sig, left, length}
 ]
```

convolution of the examples signal with the low-pass filter

In[13]:=
```
convolve[mysignal[20], lfilter];
```

In[14]:=
```
showsignal[%]
```

Out[14]=


convolution of the examples signal with the high-pass filter

In[15]:=
```
convolve[mysignal[20], hfilter];
```

In[16]:=
```
showsignal[%]
```

Out[16]=


the convolution theorem visualized

converting a finite signal into its z-transform

In[17]:=
```
sig2ser[signal_, z_] := Module[{data, left, length},
  {data, left, length} = signal;
  Sum[data[[k]] z^(k + left - 1), {k, 1, length}]
 ]
```

z-transform of the low-pass filter

In[18]:=
```
sig2ser[lfilter, z]
```

Out[18]= $\frac{1}{2} + \frac{1}{4 z} + \frac{z}{4}$

z-transform of the high-pass filter

In[19]:=
```
sig2ser[hfilter, z]
```

Out[19]= $\frac{1}{2} - \frac{1}{4 z} - \frac{z}{4}$

frequency representation of a signal

In[20]:=
```
fourier[signal_, ω_] := Module[{z},
  sig2ser[signal, z] /. z → Exp[I ω]
 ]
```

frequency representation of the low-pass-filter

In[21]:=
```
fourier[lfilter, ω]
```

Out[21]= $\frac{1}{2} + \frac{e^{-i \omega}}{4} + \frac{e^{i \omega}}{4}$

In[22]:=
```
Plot[fourier[lfilter, ω], {ω, 0, Pi}, Filling → Axis]
```

Out[22]=



frequency representation of the high-pass filter

In[23]:=
```
fourier[hfilter, ω]
```

Out[23]=
$$\frac{1}{2} - \frac{e^{-i\omega}}{4} - \frac{e^{i\omega}}{4}$$

In[24]:=
```
Plot[fourier[hfilter, ω], {ω, 0, Pi}, Filling → Axis]
```

Out[24]=


plotting the frequency representation of the example signal

In[25]:=
```
Plot[
 fourier[mysignal[20], ω],
 {ω, -Pi, Pi}, Filling → Axis, PlotRange → {-0.5, 2}]
```

Out[25]=


the product of the frequency representations of the high-pass filter and the example signal

In[26]:=

```
fourier[hfilter, ω] fourier[mysignal[20], ω]
```

Out[26]=

$$\left(\frac{1}{2}-\frac{e^{-i\omega}}{4}-\frac{e^{i\omega}}{4}\right)\left(\frac{19}{80}+\frac{11\,e^{-i\omega}}{100}+\frac{11\,e^{i\omega}}{100}+\frac{19}{125}e^{-2i\omega}+\frac{19}{125}e^{2i\omega}+\frac{44}{625}e^{-3i\omega}+\frac{44}{625}e^{3i\omega}+\frac{304\,e^{-4i\omega}}{3125}+\right.$$
$$\frac{304\,e^{4i\omega}}{3125}+\frac{704\,e^{-5i\omega}}{15\,625}+\frac{704\,e^{5i\omega}}{15\,625}+\frac{4864\,e^{-6i\omega}}{78\,125}+\frac{4864\,e^{6i\omega}}{78\,125}+\frac{11\,264\,e^{-7i\omega}}{390\,625}+$$
$$\frac{11\,264\,e^{7i\omega}}{390\,625}+\frac{77\,824\,e^{-8i\omega}}{1\,953\,125}+\frac{77\,824\,e^{8i\omega}}{1\,953\,125}+\frac{180\,224\,e^{-9i\omega}}{9\,765\,625}+\frac{180\,224\,e^{9i\omega}}{9\,765\,625}+$$
$$\frac{1\,245\,184\,e^{-10i\omega}}{48\,828\,125}+\frac{1\,245\,184\,e^{10i\omega}}{48\,828\,125}+\frac{2\,883\,584\,e^{-11i\omega}}{244\,140\,625}+\frac{2\,883\,584\,e^{11i\omega}}{244\,140\,625}+$$
$$\frac{19\,922\,944\,e^{-12i\omega}}{1\,220\,703\,125}+\frac{19\,922\,944\,e^{12i\omega}}{1\,220\,703\,125}+\frac{46\,137\,344\,e^{-13i\omega}}{6\,103\,515\,625}+\frac{46\,137\,344\,e^{13i\omega}}{6\,103\,515\,625}+$$
$$\frac{318\,767\,104\,e^{-14i\omega}}{30\,517\,578\,125}+\frac{318\,767\,104\,e^{14i\omega}}{30\,517\,578\,125}+\frac{738\,197\,504\,e^{-15i\omega}}{152\,587\,890\,625}+$$
$$\frac{738\,197\,504\,e^{15i\omega}}{152\,587\,890\,625}+\left(5\,100\,273\,664\,e^{-16i\omega}\right)/\,762\,939\,453\,125+$$
$$\frac{5\,100\,273\,664\,e^{16i\omega}}{762\,939\,453\,125}+\left(11\,811\,160\,064\,e^{-17i\omega}\right)/\,3\,814\,697\,265\,625+$$
$$\left(11\,811\,160\,064\,e^{17i\omega}\right)/\,3\,814\,697\,265\,625+$$
$$\left(81\,604\,378\,624\,e^{-18i\omega}\right)/\,19\,073\,486\,328\,125+\left(81\,604\,378\,624\,e^{18i\omega}\right)/$$
$$19\,073\,486\,328\,125+\left(188\,978\,561\,024\,e^{-19i\omega}\right)/\,95\,367\,431\,640\,625+$$
$$\left(188\,978\,561\,024\,e^{19i\omega}\right)/\,95\,367\,431\,640\,625+\left(1\,305\,670\,057\,984\,e^{-20i\omega}\right)/$$
$$\left.476\,837\,158\,203\,125+\left(1\,305\,670\,057\,984\,e^{20i\omega}\right)/\,476\,837\,158\,203\,125\right)$$

the frequency representation of the convolution of the high-pass filter and the example signal

In[27]:=
```
fourier[convolve[hfilter, mysignal[20]], ω]
```

Out[27]=

$$
\begin{aligned}
&\frac{51}{800} - \frac{339\, e^{-i\omega}}{8000} - \frac{339\, e^{i\omega}}{8000} + \frac{309\, e^{-2 i\omega}}{10\,000} + \frac{309\, e^{2 i\omega}}{10\,000} - \frac{339\, e^{-3 i\omega}}{12\,500} - \frac{339\, e^{3 i\omega}}{12\,500} + \frac{309\, e^{-4 i\omega}}{15\,625} + \\
&\frac{309\, e^{4 i\omega}}{15\,625} - \frac{1356\, e^{-5 i\omega}}{78\,125} - \frac{1356\, e^{5 i\omega}}{78\,125} + \frac{4944\, e^{-6 i\omega}}{390\,625} + \frac{4944\, e^{6 i\omega}}{390\,625} - \frac{21\,696\, e^{-7 i\omega}}{1\,953\,125} - \\
&\frac{21\,696\, e^{7 i\omega}}{1\,953\,125} + \frac{79\,104\, e^{-8 i\omega}}{9\,765\,625} + \frac{79\,104\, e^{8 i\omega}}{9\,765\,625} - \frac{347\,136\, e^{-9 i\omega}}{48\,828\,125} - \frac{347\,136\, e^{9 i\omega}}{48\,828\,125} + \\
&\frac{1\,265\,664\, e^{-10 i\omega}}{244\,140\,625} + \frac{1\,265\,664\, e^{10 i\omega}}{244\,140\,625} - \frac{5\,554\,176\, e^{-11 i\omega}}{1\,220\,703\,125} - \frac{5\,554\,176\, e^{11 i\omega}}{1\,220\,703\,125} + \\
&\frac{20\,250\,624\, e^{-12 i\omega}}{6\,103\,515\,625} + \frac{20\,250\,624\, e^{12 i\omega}}{6\,103\,515\,625} - \frac{88\,866\,816\, e^{-13 i\omega}}{30\,517\,578\,125} - \frac{88\,866\,816\, e^{13 i\omega}}{30\,517\,578\,125} + \\
&\frac{324\,009\,984\, e^{-14 i\omega}}{152\,587\,890\,625} + \frac{324\,009\,984\, e^{14 i\omega}}{152\,587\,890\,625} - \frac{1\,421\,869\,056\, e^{-15 i\omega}}{762\,939\,453\,125} - \frac{1\,421\,869\,056\, e^{15 i\omega}}{762\,939\,453\,125} + \\
&\frac{5\,184\,159\,744\, e^{-16 i\omega}}{3\,814\,697\,265\,625} + \frac{5\,184\,159\,744\, e^{16 i\omega}}{3\,814\,697\,265\,625} - \frac{22\,749\,904\,896\, e^{-17 i\omega}}{19\,073\,486\,328\,125} - \\
&\frac{22\,749\,904\,896\, e^{17 i\omega}}{19\,073\,486\,328\,125} + \frac{82\,946\,555\,904\, e^{-18 i\omega}}{95\,367\,431\,640\,625} + \frac{82\,946\,555\,904\, e^{18 i\omega}}{95\,367\,431\,640\,625} - \\
&\left(363\,998\,478\,336\, e^{-19 i\omega}\right) / 476\,837\,158\,203\,125 - \frac{363\,998\,478\,336\, e^{19 i\omega}}{476\,837\,158\,203\,125} + \\
&\left(416\,611\,827\,712\, e^{-20 i\omega}\right) / 476\,837\,158\,203\,125 + \frac{416\,611\,827\,712\, e^{20 i\omega}}{476\,837\,158\,203\,125} - \\
&\left(326\,417\,514\,496\, e^{-21 i\omega}\right) / 476\,837\,158\,203\,125 - \frac{326\,417\,514\,496\, e^{21 i\omega}}{476\,837\,158\,203\,125}
\end{aligned}
$$

checking the convolution theorem

In[28]:=
```
Simplify[%% - %]
```

Out[28]=
```
0
```

the effect of the low-pass filter

In[29]:=

```
Plot[{
  fourier[mysignal[40], ω] * fourier[lfilter, ω],
  fourier[mysignal[40], ω],
  fourier[lfilter, ω]},
 {ω, -Pi, Pi}, PlotRange → All, Filling → Axis]
```

Out[29]=



the effect of the high-pass filter

In[30]:=

```
Plot[{
  fourier[mysignal[40], ω] * fourier[hfilter, ω], fourier[mysignal[40], ω],
  fourier[hfilter, ω]},
 {ω, -Pi, Pi}, PlotRange → All, Filling → Axis]
```

Out[30]=